

# Neo4j

Mimanshu Maheshwari

October 4, 2024

# Contents

0.1	Setting up Neo4j with docker . . . . .	3
0.2	Creating Nodes . . . . .	3
0.3	Creating Relationships . . . . .	4
0.4	Defining a constraint . . . . .	4
0.5	MATCH . . . . .	4
0.6	Optional Match . . . . .	5
0.7	WHERE . . . . .	5
0.8	Merge . . . . .	5
0.9	UNION, UNION ALL . . . . .	5
0.10	ORDER BY, SKIP, LIMIT, DISTINCT . . . . .	5
0.11	SET . . . . .	6
0.12	FOREACH . . . . .	6
0.13	REMOVE . . . . .	6
0.14	DELETE . . . . .	6
0.15	DETACH DELETE . . . . .	7
0.16	Bulk Load of Data . . . . .	7
0.17	Functions . . . . .	8
0.17.1	toLower() . . . . .	8
0.17.2	toUpper() . . . . .	8
0.17.3	substring() . . . . .	8
0.17.4	replase() . . . . .	8
0.17.5	reverse() . . . . .	8
0.17.6	Aggregate function . . . . .	8
0.17.7	List functions . . . . .	9
0.18	User defined functions(UDF) in Neo4J . . . . .	9
0.18.1	UDFs in Neo4J . . . . .	9
0.18.2	Deploying UDF . . . . .	10
0.18.3	Calling a UDF . . . . .	10
0.19	Optional Schema Indexes and constraints . . . . .	10
0.19.1	Analyze query performance using EXPLAIN . . . . .	10
0.19.2	Improve performance using indexes . . . . .	10
0.20	Working with constraints . . . . .	11
0.20.1	Unique node property constraint . . . . .	11
0.21	Using Neo4J with Java . . . . .	12

# List of Figures

# Listings

1	get neo4j image . . . . .	3
2	Starting neo4j container . . . . .	3
3	Create Node . . . . .	3
4	Create Node Example . . . . .	4
5	Create Relationships . . . . .	4
6	Create Relation Example . . . . .	4
7	constraint . . . . .	4
8	Match syntax . . . . .	4
9	Match example . . . . .	4
10	Match example . . . . .	4
11	Match example . . . . .	5
12	Match example . . . . .	6
13	Neo4J with java . . . . .	12

## 0.1 Setting up Neo4j with docker

Using official docker setup [link](#) for neo4j.

```
1 docker pull neo4j
```

Listing 1: get neo4j image

```
1 docker run \  
2 --publish=7474:7474 --publish=7687:7687 \  
3 --volume=$HOME/neo4j/data:/data \  
4 neo4j
```

Listing 2: Starting neo4j container

## 0.2 Creating Nodes

```
1 CREATE( var: label{key1:value1, key2:value2, key3:value3,...,keyn:valuen})
```

Listing 3: Create Node

- Properties of nodes present between { and }
- Node present between *and*
  1. var is the variable name such as John
  2. label is the entity type such as Customer

3. key:value are the attributes of the properties such as `contact_num: 1234567890`

Example:

```
1 CREATE (b:Bank{name:"Oziku"})
```

Listing 4: Create Node Example

## 0.3 Creating Relationships

```
1 CREATE (node1) -[var:Rel_type{key1:value1, key2:value2, ...keyn:valuen}]->(node2)
```

Listing 5: Create Relationships

Example:

```
1 CREATE (BG:Customer {cust_id: 675489, cust_name: "BlackGreyTechnologies",  
    ↪ contact_num: 1298764592 })-  
2 [:Owns]->(AcBG:Account {acc_num: 3498761, cust_id: 675489, type: "checking",  
    ↪ balance:958990 })
```

Listing 6: Create Relation Example

Nodes, relationships can be created with/without label, properties Nodes or relationships can have single or multiple labels

## 0.4 Defining a constraint

Create a unique constraint on `cust_id` property of the Customer node as shown below:

```
1 CREATE CONSTRAINT ON (c:Customer) ASSERT c.cust_id IS UNIQUE
```

Listing 7: constraint

## 0.5 MATCH

Is used to search for a pattern. To return all the nodes created:

```
1 MATCH (n) RETURN n
```

Listing 8: Match syntax

```
1 MATCH (n:Bank)  
2 RETURN n
```

Listing 9: Match example

```
1 MATCH (Customer {cust_name: 'BlackGreyTechnologies' })--(Account)  
2 RETURN Account.acc_num, Account.balance
```

Listing 10: Match example

```

1 MATCH (city:City {name:"Raleigh"})
2 MERGE (state:State{name:"North Carolina"})
3 MERGE (city)-[:LOCATED_IN]->(state)
4 RETURN city,state

```

Listing 11: Match example

## 0.6 Optional Match

```

1 MATCH (c:Customer {cust_id: 675489 })
2 OPTIONAL MATCH (c)-[r:having] -()
3 RETURN c.cust_name, r.name

```

## 0.7 WHERE

```

1 MATCH (c:Customer), (a:Account)
2 WHERE a.type='checking'
3 RETURN c.cust_name,a.acc_num,a.balance

```

## 0.8 Merge

```

1 MERGE (c:Customer {cust_name: 'Charlie Sheen' })
2 RETURN c

```

## 0.9 UNION, UNION ALL

Union combines the results of multiple queries and removes duplicates whereas Union All performs the same operation but retains duplicates.

```

1 MATCH (c:Customer), (a:Account)
2 WHERE a.type='checking'
3 RETURN c.cust_name,a.acc_num,a.balance
4 UNION
5 MATCH (c:Customer {cust_name: 'BlackGreyTechnologies' })--(a:Account)
6 RETURN c.cust_name, a.acc_num, a.balance

```

NOTE: All sub queries must have the same column names.

## 0.10 ORDER BY, SKIP, LIMIT, DISTINCT

ORDER BY - specifies how the output of RETURN or WITH should be sorted. SKIP - defines from which record to start including the records in the output. LIMIT - constraints the number of output records. DISTINCT - retrieves only unique rows. The below query sorts the Customer node in the ascending order of its customer names, skips the first record and limits the output to only one record:

```
1 ORDER BY, SKIP, LIMIT, DISTINCT
```

To retrieve the unique relationships present in the database:

```
1 MATCH (n)-[r]-()
2 RETURN distinct type(r)
```

## 0.11 SET

It is used to update node labels and properties of nodes and relationships. For example, adding new properties such as email and country to the existing Customer BlackGreyTechnologies:

```
1 MATCH (c:Customer {cust_name: 'BlackGreyTechnologies'})
2 SET c.email= 'BGT@blackgrey.com', c.country ='France'
3 RETURN c
```

Listing 12: Match example

## 0.12 FOREACH

It updates data within a list which can be components of a path\* or result of an aggregation Path is a directed sequence of nodes and relationships. Assume you want to track funds transfer between two accounts suspected to be laundering funds illegally between intermediary accounts.

```
1 CREATE p =(:Account {acc_num:65178})-[:Funds_transfer]->(:Account
2 {acc_num:98567})-[:Fundstransfer]->(:Account {acc_num:46378})-[:Fundstransfer]->(:
   ↳ Account
3 {acc_num:95648})-[:Fundstransfer]->(:Account {acc_num:46897})
4 RETURN p
5
6 MATCH p =(:Account {acc_num:65178})-[*]->(:Account {acc_num:46897})
7 FOREACH (n IN nodes(p) | SET n.marked ="flaggedFraud")
```

Note: In the above code, [\*] is used to define any number of intermediary relationships acc\_num=65178 to ↳ acc\_num=46897 In the output shown below, you can observe that a new property marked:"flaggedFraud" has been added:

## 0.13 REMOVE

Is used to remove labels and properties of nodes and relationships.

To remove the email property from Customer having cust\_name=BlackGreyTechnologies:

```
1 MATCH (c:Customer {cust_name: 'BlackGreyTechnologies'})
2 REMOVE c.email
3 RETURN c
```

## 0.14 DELETE

Is used to delete nodes, relationships or paths. Node cannot be deleted without deleting its associated relationships. Either delete relationships explicitly or use DETACH DELETE that is discussed next.

```

1 MATCH (n {acc_num: 65178 })-[r:Funds_transfer]->()
2 DELETE r

```

## 0.15 DETACH DELETE

Is used to delete nodes along with their relationships.

To delete Customer node with name: "BlackGreyTechnologies" and all its associated links:

```

1 MATCH (a {cust_name: "BlackGreyTechnologies"})
2 DETACH DELETE a

```

Delete all the nodes and relationships from the database:

```

1 MATCH (n)
2 OPTIONAL MATCH (n)-[r]-()
3 DELETE n,r

```

## 0.16 Bulk Load of Data

Step 1: To load data from csv file, the below configuration property needs to be added to the neo4j.conf configuration file: dbms.security.allow\_csv\_import\_from\_file\_urls=true

Step 2: Create Bank node, Customer node and ensure that the customer ID is unique as discussed previously.

Step 3: Load data from Customer.csv file to Customer node

```

1 load csv with headers from 'file:///C:/Users/Sahana_Basavaraja/Desktop/Customer.csv'
  ↪ as cust
2 merge (c:Customer{cust_id:toInteger(cust.cid),cust_name:cust.cname,contact_num:
  ↪ toInteger(cust.phnum)})

```

Note: By default, the row retrieved from the file is always string, for example cust. You need to explicitly convert to the appropriate datatypes if required, for example toInteger() as shown above. Establish the relationship between Customer and Bank as Customer\_of:

```

1 MATCH (c:Customer),(b:Bank) merge (c)-[b1:Customer_of]->(b)
2 MATCH (n) RETURN n
3 CREATE CONSTRAINT ON (a:Account) ASSERT a.acc_num IS UNIQUE
4
5 // load data present in account csv
6 load csv with headers from 'file:///C:/Users/Sahana_Basavaraja/Desktop/Account.csv'
  ↪ as acc
7 merge (a:Account{acc_num:toInteger(acc.acc_num)})
8 set
9 a.cust_id=toInteger(acc.cid),
10 a.balance=toInteger(acc.balance)

```

Note: Use SET to map columns of the file to properties of the node as illustrated above. Establish the relationship between Customer and Account as Owns:

```

1 load csv with headers from 'file:///C:/Users/Sahana_Basavaraja/Desktop/Account.csv'
  ↪ as acc

```



```

2 match (a:Account{acc_num:toInteger(acc.acc_num)}), (c:Customer{cust_id:toInteger(acc.
  ↳ cid)})
3 merge (a) <-[o:Owns]-(c)

```

```

1 load csv with headers from 'file:///C:/Users/Sahana_Basavaraja/Desktop/txn.csv' as
  ↳ txn
2 match (a:Account), (b:Account)
3 where a.acc_num=toInteger(txn.from_acc) AND b.acc_num=toInteger(txn.to_acc) AND
  ↳ toInteger(a.balance)>=toInteger(txn.amount_acc)
4 create (a)-[:Funds_transfer{txn_id:toInteger(txn.tid),from_acc:toInteger(txn.
  ↳ from_acc),to_acc:toInteger(txn.to_acc),amount:toInteger
5 (txn.amount_acc),txntime:txn.txntime}]>-(b)

```

## 0.17 Functions

### 0.17.1 toLower()

```

1 MATCH (c:Customer) RETURN toLower(c.cust_name) LIMIT 1

```

### 0.17.2 toUpper()

```

1 MATCH (c:Customer) RETURN toUpper(c.cust_name) LIMIT 1

```

### 0.17.3 substring()

```

1 MATCH (c:Customer) RETURN SUBSTRING(c.cust_name,0,5) LIMIT 1

```

### 0.17.4 replace()

```

1 MATCH (c:Customer{cust_name:"BlackGreyTechnologies"}) RETURN replace("
  ↳ BlackGreyTechnologies","Grey","Gradient")

```

### 0.17.5 reverse()

```

1 MATCH (c:Customer) RETURN reverse(c.cust_name) LIMIT 1

```

### 0.17.6 Aggregate function

It is similar to GROUP BY clause in SQL. It takes multiple values as arguments, performs computation on it and returns the computed value.

- sum() - returns the sum of a set of values
- avg() - returns the average of a set of values

- `count()` - returns number of rows or values
- `max()` - returns maximum value in set of values
- `min()` - returns minimum value in set of values

The frauds do not transfer amount in one single transaction. Multiple transactions would take place in order to hide the transit of illegally obtained money.

The below query retrieves the details of the account number having highest `total_amount` transferred in a series that is greater than 20K:

```
1 MATCH (a)-[r:Funds_transfer]->(b)
2 WITH r.to_acc AS account_number,
3 count(r.to_acc) as total_txn_to_acc,
4 sum(r.amount) as total_amount
5 WHERE total_amount >20000
6 RETURN account_number,total_amount,total_txn_to_acc
7 ORDER BY total_amount DESC
8 LIMIT 1
```

### 0.17.7 List functions

It returns a list of nodes and relationships in a path, labels, keys.

- `keys()` - returns a list of property names of a node, relationship or map
- `labels()` - returns a list of all the labels of a node
- `nodes()` - returns a list of all nodes in a path
- `relationships()` - returns a list of all relationships in a path

To retrieve the list of distinct labels and its respective property keys:

```
1 MATCH (n)
2 RETURN labels(n), keys(n)
```

## 0.18 User defined functions(UDF) in Neo4J

Neo4J has many built-in functions. In order to extend the functionalities of Neo4J, you can create your own UDFs in Java.

### 0.18.1 UDFs in Neo4J

- are read-only and always returns a single-value
- is annotated with `@UserFunction`
- valid input types and output types are `string`, `long`, `double`, `boolean`, `node`, `relationship`, `path`,  
↔ `object`, `map<K(string),V>`, `list`
- is called as `package-name.method-name`

From the Account node, retrieve the average balance amount.

```

1 package com.infy.bda;
2 import java.util.List;
3 import org.neo4j.procedure.Description;
4 import org.neo4j.procedure.Name;
5 import org.neo4j.procedure.UserFunction;
6 public class AverageBalance {
7     @UserFunction
8     @Description("com.infy.bda.AverageBalance([0.5,1,2.3]) returns the average of the given list
9         ↪ of values")
10    public double avg(@Name("numbers") List<Number> list) {
11        double avg = 0;
12        for (Number number : list) {
13            avg += number.doubleValue();
14        }
15        return (avg/(double)list.size());
16    }

```

## 0.18.2 Deploying UDF

UDFs are packaged in a jar file. You have to copy the jar file into the plugins directory of your Neo4J server and restart.

## 0.18.3 Calling a UDF

Before calling the UDF, to see the list of UDFs deployed in your server, use the following command:

```
1 CALL dbms.functions()
```

User-defined functions are called in the same way as any other Cypher function. The function name must be fully qualified, that is, a function named avg defined in the package com.infy.bda is called using:

```

1 MATCH (a:Account)
2 RETURN com.infy.bda.avg(collect(a.balance))

```

## 0.19 Optional Schema Indexes and constraints

### 0.19.1 Analyze query performance using EXPLAIN

Query performance can be analysed using the EXPLAIN clause. It provides the execution plan for the Cypher query. Consider the below query that retrieves the Customer node having cust\_name:"George Clooney":

```

1 EXPLAIN
2 MATCH (c:Customer{cust_name:"George Clooney"})
3 RETURN c

```

Few properties generated from the output of EXPLAIN command are explained below:

- NodeByLabelScan: Scanning the nodes based on a specific label
- Filter: Conditional expression

### 0.19.2 Improve performance using indexes

Create Index:

```

1 CREATE INDEX ON :Label(property) //Index on single property
2 CREATE INDEX ON :Label(property1, property2, ...,propertyn) //Index on multiple
  ↪ properties

```

Example:

```

1 CREATE INDEX ON :Customer(cust_name)

```

**View Index:**

```

1 CALL db.indexes()

```

**Drop Index:**

```

1 DROP INDEX ON :Label(property)
2 DROP INDEX ON :Label(property1, property2, .....propertyn)

```

Example:

```

1 DROP INDEX ON :Customer(cust_name)

```

## 0.20 Working with constraints

Data integrity in Neo4J can be enforced using constraints. Two types of constraints can be created:

- Unique node property constraint
- Property and relationship existence constraint<sup>1</sup>

### 0.20.1 Unique node property constraint

It ensures that property values for all nodes corresponding to a particular label are unique.

#### Creation of unique node constraint

Earlier in the course, you have created a unique constraint on the `cust_id` property of the `Customer` node as shown below:

```

1 CREATE CONSTRAINT ON (c:Customer) ASSERT c.cust_id IS UNIQUE

```

NOTE:

- Nodes without the property used in the constraint will have no effect
- Creation of unique property constraint will also create index on that property

<sup>1</sup>Available only in the Neo4J Enterprise Edition.

## Drop unique constraint:

Use DROP CONSTRAINT clause to remove a constraint from the database as shown below:

```
1 DROP CONSTRAINT ON (c:Customer) ASSERT c.cust_id IS UNIQUE
```

Note: Dropping the constraint will also remove the index created.

## 0.21 Using Neo4J with Java

Neo4J provides developer kits for various languages like Java, .NET, Python, PHP, Go, C and so on. Below code can be used to access Neo4J server from Java application using Neo4J's binary Bolt<sup>2</sup> protocol. You can copy paste the following code in Eclipse and include [neo4j](#) and [neo4j-java driver](#) jars.

```
1 package com.infy;
2
3 import org.neo4j.driver.v1.AccessMode;
4 import org.neo4j.driver.v1.AuthTokens;
5 import org.neo4j.driver.v1.Driver;
6 import org.neo4j.driver.v1.GraphDatabase;
7 import org.neo4j.driver.v1.Session;
8 import org.neo4j.driver.v1.StatementResult;
9 import org.neo4j.driver.v1.Transaction;
10 import static org.neo4j.driver.v1.Values.parameters;
11 import java.util.ArrayList;
12 import java.util.List;
13
14 public class NeoJavaDemo implements AutoCloseable {
15     private Driver driver;
16
17     public NeoJavaDemo( String uri, String user, String password ) {
18         this.driver = GraphDatabase.driver( uri, AuthTokens.basic( user, password ) );
19     }
20
21     @Override
22     public void close() throws Exception {
23         this.driver.close();
24     }
25
26     // Create a Bank node
27     private StatementResult addBank( final Transaction tx, final String name ) {
28         return tx.run( "CREATE (:Bank {name: $name})", parameters( "name", name ) );
29     }
30
31     // Create a Customer node
32     private StatementResult addCustomer( final Transaction tx, final String cust_name ) {
33         return tx.run( "CREATE (:Customer {cust_name: $cust_name})", parameters( "cust_name",
34             ↪ cust_name ) );
35
36     // Create a customer_of relationship to a pre-existing bank node.
37     // This relies on the customer first having been created.
38     private StatementResult cust( final Transaction tx, final String customer, final String bank )
39         ↪ {
40         return tx.run( "MATCH (c:Customer {cust_name: $cust_name}) " +
41             "MATCH (b:Bank {name: $name}) " +
42             "CREATE (c)-[:Customer_of]->(b)",
43             parameters( "cust_name", customer, "name", bank ) );
44     }
45
46     public void addCustomerAndBank() {
47         // To collect the session bookmarks
```

<sup>2</sup>Bolt protocol is a connection oriented network protocol used for client-server communication in database applications. By default, it runs on 7687 port.

```

47     List<String> savedBookmarks = new ArrayList<>();
48     // Create the first customer and bank relationship.
49     try ( Session session1 = this.driver.session( AccessMode.WRITE ) ) {
50         session1.writeTransaction( tx -> addBank( tx, "Oziku" ) );
51         session1.writeTransaction( tx -> addCustomer( tx, "George" ) );
52         session1.writeTransaction( tx -> cust( tx, "George", "Oziku" ) );
53         savedBookmarks.add( session1.lastBookmark() );
54     }
55 }
56
57 public static void main( String[] args ) throws Exception {
58     try ( NeoJavaDemo g = new NeoJavaDemo( "bolt://localhost:7687", "neo4j", "Infy@123#" ) ) {
59         g.addCustomerAndBank();
60     }
61 }
62 }

```

Listing 13: Neo4J with java

- The above Neo4J client connects to the Neo4J server running on localhost with neo4j and Infy@123# as username and password respectively.
- To connect to a remote Neo4J server, replace localhost with the server's hostname and provide relevant username and password.

For more details view [Neo4J Drivers](#).